
Spike2 and Signal Training Days
Our US Spike2 and Signal training courses will be held on Thursday 1* and Friday 2" of
November, prior to the Society for Neuroscience meeting in San Diego. The venue will be
the Holiday Inn San Diego — On The Bay. These training days are for both existing and
prospective users of Spike2 and Signal and are suitable for beginners and advanced .
users. There is also our popular Workshop thread where you can discuss your particular
requirements with a CED engineer or programmer in a small group environment.
If you or any of your colleagues would like to attend, please follow the links to view the
training day program and complete the registration form. Please give a short description -_
of the type of research that you do and analysis that you are interested in. This will allow
us to tailor sessions for specific requirements.
Meetings and events
Scandinavian Physiological Society 2007
University of Oslo
Oslo, Norway
August 10" — 12" 2007
Scottish Neuroscience Group Annual Meeting
University of Edinburgh
Edinburgh
August 31* 2007
IX meeting of the International Basal Ganglia Society (IBAGS)
Hotel Zuiderdein
Egmond aan Zee, Netherlands
September 2™ — 6" 2007
Cardiac Electrophysiology and EWGCCE
University of Manchester
Manchester
September 5™ — 8" 2007
European Society for Muscle Research - EMC
Essinge Conference Center
Stockholm
September 8" — 12" 2007
Latest versions of Spike2 and Signal
Updates for Spike2 and Signal are available from the CED downloads page, or by
clicking on the links in the table below. Demonstration versions of the latest software are
also available.
Spike2 downloads Signal downloads
Spike?2 version 6.04 Signal version 3.09 ¢
Spike?2 version 5.18 Signal version 2.16 ‘
Spike2 demo Signal demo .



http://www.ced.co.uk/nw9y.htm
http://www.ced.co.uk/tdapply.htm
http://www.imbv.uio.no/SPS2007/
http://www.edinburghneuroscience.ed.ac.uk/SNG2007meeting.html
http://www.ibags.info/index.php
http://www.physoc.org/meetings/man2007.asp
http://www.esmr.org/EMCSth1.htm
http://www.ced.co.uk/uplsu.shtml
http://www.ced.co.uk/s2wu6u.shtml
http://www.ced.co.uk/s2wu5u.shtml
http://www.ced.co.uk/s2demu.shtml
http://www.ced.co.uk/sigw3u.shtml
http://www.ced.co.uk/sigw2u.shtml
http://www.ced.co.uk/sigdemu.shtml

| have now spent some time with Active Cursors after reading the last
eNewsletter and can detect the features of interest in my recorded ECG trace.
Now | would like to start taking measurements based on my feature detection.

Spike2 can generate automated measurements on or off-line based on cursors
positioned by the results of feature detection as demonstrated in the previous
eNewsletter (click here to view eNewsletter #28 on our website). There are two
ways of taking automated measurements in Spike2. They can be aligned in time
with the raw data by plotting results to new channels in the file, or they can be
plotted to a separate XY scatter view. An XY view can process multiple
measurement channels simultaneously and the results can be easily exported to
spreadsheet programs using the Copy as Text command from the Edit menu.

Having already set-up our Active Cursors in our ECG trace in the previous
eNewsletter, we could now use the Measurements - XY view function from the
Analysis menu to automatically step Cursor O through the data finding every R-
wave peak and updating the position of the other Cursors in response.
Measurements such as mean waveform value, slope and time difference, as well
as many others, can then be taken at every step and the results plotted to the XY
view.

The example below is set up to process a measurement of the Q-R interval
based on the cursor positions in our example ECG data. The Settings dialog
includes a section that defines the stepping mode for Cursor 0 (which uses the
current Active Mode for the cursor if one is already set) as well as the type and
time of each X and Y measurement value.

You can define the titles for the Plot Channel as above by simply editing the text
in the dialog field. Our X Measurement takes the time at point of Cursor O for
each step (the time at which each R wave occurs in the data file). Our Y
measurement takes the time difference between Cursor 2 (Q wave) and Cursor 0
(R wave) for every step through the data. Pressing New creates the XY view and
opens a new dialog that sets the time range in the source data from which to take
measurements. Taking measurements from 0 to the maximum time (MaxTime())
in our source file gives the plot below.

Hint
When plotting measurements to a data channel, you can make use of the different channel draw modes for marker and event data. This means that you could display the measurement time points as mean frequencey to give a display of heart rate, for example. See eNewsletter 6 for more information.

http://www.ced.co.uk/img/cednws28.pdf
http://www.ced.co.uk/img/cednws28.pdf

Trend plot of Q-R Interval processed from 0 to MaxTime()

By right-clicking on the trend plot we can make changes to the settings and add a
second Plot Channel that will measure the time difference between Cursor 0 (R
wave) and Cursor 1 (S wave) for every step, as below.

XY plot containing the results of 2 separate measurements

XY views can contain up to 256 Plot Channels holding the results of different
measurements. For more information on using the measurement functions in
Spike2, see the manual or on-line help.

Signal

Q.

>

I would like to be able to introduce random intervals of 5 to 10 seconds between
sampling sweeps. Would this require a script?

This functionality is already available in Signal when using Fixed Interval sweep
mode, which can be selected from the General tab of the sampling configuration.
The parameters to determine the intervals between sweeps are then set within
the Pulses configuration dialog, which can be found in the Outputs tab of the
sampling configuration. With Fixed Interval mode selected, two additional fields
are enabled in the dialog, one to set the default interval between sweeps and a
Vary % field that can be used to add a random variation to the interval time.

Hint
Sweeps in Fixed Interval mode are timed internally by the 1401 so that they occur at the specified interval. For this reason, external triggers are not used in this sampling mode


For this example, an Interval setting of 7.5 (s) and a Vary % setting of 33 would
give a random interval of 5 to 10 seconds between each sampled sweep as 33%
of 7.5 is 2.5, giving a random time from 7.5-2.5 to 7.5+2.5 seconds.

Pulse configuration dialog when using Fixed Interval mode

Scripts: Spike2

Q.

I will be ma mechanical stimulator, controlled by a voltage ramp, to evoke a
response from my skin-nerve prep. | know that | can use the output sequencer to
generate the ramp, but | would like to also like to be able to set the maximum
ramp threshold once a certain number of spikes have been evoked and then use
this ramp threshold level for the rest of the experiment. Is there a script or
sequence available that would give me this kind of on-line control?

The attached script, RampThresh.s2s and the associated sequence file,
RampThreshSeq.pls, demonstrates a method of updating stimulus parameters
in response to activity on a waveform channel. In this example a voltage ramp is
used to drive a stimulator while a selected channel is monitored for spike activity.
The threshold for spike detection is defined by the amplitude difference between
horizontal cursors that are positioned by the user on the selected spike data
channel. Once the script has detected a pre-defined number of spikes, it stops
the ramp and sets the amplitude level of the ramp at that time as the upper limit
for the ramp stimulus for the rest of the experiment.

For testing purposes, connect DAC 0 to ADC port 0 on the 1401 and set up a
suitable sampling configuration to visualize the ramp behaviour and load the
sequence file into the configuration before running the script. The script
generates a toolbar that is used to control sampling and set spike monitoring
parameters. The toolbar buttons and dialog fields contain tooltips that explain
their functions when you ‘hover’ the mouse pointer over them. The Get
Threshold button is enabled once sampling has started and is used to generate
ramp output and monitor the spike channel for activity. Once the set number of
spikes has occurred, the ramp is halted. Once the upper limit for the ramp has
been defined in this way, the Stimulate button can be used to repeat the ramp
stimulus for the remainder of the experiment.

Threshold ramp halted by a single detected spike, followed by two rapid stimulus
ramps that then use the same amplitude level

Right-click the script icon
and save to disk.

If you have any problems
opening the embedded
scripts in this newsletter
please let us know.



'$RampThresh.s2s|Spike2 script to control mechanical stimulator and set ramp peak

'in response to spike activity



'The script is a 'work in progress' and is offered without guarantees.

'You must test it to see whether it is suitable for your application.



'CED 06/07/2007



var data%;                  'view handle of data file

var spChan% := 5;           'Spike channels

var nSPikes% := 1;          'no. of spikes to detect

var spTime := 1;            'time (in secs)

var threshAmp%;              'Stimulus level at threshold

var nChan%;                 'Memory channel to contain spikes

var sTime,eTime;           'current and last time in Idle routine

var spCount%;               'Count of spikes

var threshFlag%;           'flag if threshold crossed

var rTime;                  'Ramp up and down time for repeated stimulus

var reps%;                  'Number of stimulus repetitions

var hTime;                  'Time to hold level for each stimulus

var int,int%;                    'interval between stimuli

var riseRate := 200;       'Rate of ramp 

var riseRate%;              'value to write to sequence

var hTime%;                 'Hold time for ramp



ToolbarSet(1, "Quit||Quit the script");      'Toolbar buttons for script control

ToolbarSet(3, "New file||Open a new file for sampling and set parameters", New%);

ToolbarSet(5, "Start||Start sampling", Start%);

ToolbarSet(7, "Get Threshold||Start ramp output and monitor spike channel",GetThresh%);

ToolbarSet(9, "Zero Output||Ramp output down to 0",Zero%);

ToolbarSet(11, "Stimulate||Generate voltage ramp using defined amplitude threshold",Go%);

ToolbarEnable(5,0);

ToolbarEnable(7,0);

ToolbarEnable(9,0);

ToolbarEnable(11,0);

Toolbar("",1023);





Func New%()

var range;

var ok%;

data% := FileNew(0,2);      'New file based on current sampling config.

FrontView(data%);           'Make sure it's current

Window(0,0,100,100);        'Make full screen

ToolbarEnable(3,0);         

ToolbarEnable(5,1);

DlgCreate("Threshold detection");       'Dialog for detedtion and ramp settings

DlgChan(1, "Spike channel|Waveform channel to monitor for spikes",1);

DlgInteger(2, "Spikes to indicate threshold|No. of spikes needed to stop ramp output",1,100);

DlgReal(3, "Spike lookback time (s)|Time range to count spikes back from current time ",0.001,60);

DlgInteger(4, "Number of stimulus repetitions|No. of ramp outputs to Stimulate",1,10000);

DlgReal(5, "Rise rate (mN/s)|Rate of ramping over time",1,1000);

DlgReal(6, "Stimulus hold time (s)|Ramp and then hold for n (s)",0.1,1000);

DlgReal(7, "interval between stimuli (s)|Interval between ramp stimuli",1,3600);

ok% := DlgShow(spChan%,nSpikes%,spTime,reps%,riseRate,hTime,int);       'Return the dialog values

riseRate% := ((riseRate/1000.0)* 2147483647)/10000;

hTime% := (hTime*10000)-2;

int% := (int*10000)-2;          'Calculate rise time and interval for write to sequencer

SampleSeqVar(2,reps%);

SampleSeqVar(3,hTime%);

SampleSeqVar(4,riseRate%);

SampleSeqVar(5,int%);           'Write values to the sequencer

HCursorDelete(-1);

range := YHigh(spChan%)-YLow(spChan%);

HCursorNew(spChan%,YLow(spChan%) + (range *2/3));   'Display horizontal cursors

HCursorNew(spChan%,YLow(spChan%) + (range/3));

nChan% := MemChan(3);

ChanShow(nChan%);

return 1;

end;





Func Start%()           'Function to start sampling

View(data%);

threshFlag% := 0;       'Set threshold flag to 0

SampleStart();              'Start sampling

ToolbarSet(5,"Stop", Stop%);        'Set a Stop button

Yield(spTime,1023);

ToolbarSet(0,"Idle",Idle%);     'Start Idle routine

ToolbarEnable(7,1);

ToolbarEnable(9,1);

return 1;

end;



Func Stop%();       'Function to Stop samplinh

View(data%);

SampleStop();       'Stop sampling

ToolbarSet(5, "Start", Start%);     'Set Start button for next time

ToolbarEnable(5,0); 

ToolbarEnable(3,1);

return 1;

end;



Func GetThresh%()

View(data%);

SampleKey("t");         'Run key t of sequence

threshFlag% := 1;      'Set flag to 1 

return 1;   

end;





Func Idle%()

View(data%);

sTime := eTime;         'Set current time

eTime := MaxTime();     'Get current time

MemImport(nChan%,spChan%,eTime-0.1,eTime,0,0,HCursor(1)-HCursor(2));    'Import data above thresholds

if threshFlag% > 0 then     'If flag for threshold is set..

    spCount% := Count(nChan%,eTime-spTime,eTime);       'Count spikes

    if spCount% >= nSpikes% then        'If no. of spikes reached

        SampleKey("h");                     'Run key h (stop ramping)

        Yield(0.5,1023);                    'Yield

        threshAmp% := SampleSeqVar(1);      'Set threshold amplitude for subsequent ramps

        threshFlag% := 0;               'set flag

        ToolbarEnable(11,1);

    endif;

endif;

return 1;

end;





Func Zero%()

View(data%);

SampleKey("z");     'Zero the ramp back to 0

return 1;

end;





Func Go%()

View(data%);

SampleKey("s");     'Run stimulus ramps (with threshold set)

return 1;

end;

david

Scripts: Signal

Q. Is it poss@to increase or decrease the sample rate in a Signal data file once
recorded? | have inadvertently recorded data files using a colleagues sampling
configuration, and would like to change the sample rate to match my previously
recorded files

A. The attached script, ReSample.sgs, can be used to interpolate a Signal data file
to increase or decrease the number of points and hence the sample rate using
cubic splining. The script creates a new copy of the file with the selected re-
sample factor.

Did you know...?
The channel numbers in a Spike2 data file are now displayed in red if a waveform

channel is modified by a channel process (rectification, smoothing etc.) or a marker
channel has a marker filter applied.

Recent questions

Q. Is there any@ to display the current time into sampling in a separate window
or toolbar, rather than just on the right hand edge of the X-axis?

A. The attached script, SampleTime.s2s, creates a ‘stopwatch’ timer in the script
toolbar that starts when sampling begins, and displays the time elapsed in hours,
minutes and seconds. This allows the user to examine previous sections of data
in the file, while still being able to note the current time elapsed since sampling
began.

User group

If you have any comments about the newsletter format and content, or wish to un-
subscribe from the mailing list for this newsletter, please notify sales@ced.co.uk.

Right-click the script icon
and save to disk.

If you have any problems
opening the embedded
scripts in this newsletter
please let us know.

'$ReSample.sgs|Signal script to produce a new re-sampled version of a Signal data file using spline interpolation



'The script is a 'work in progress' and is offered without guarantees.

'You must test it to see whether it is suitable for your application.



'CED 06/07/2007





var oldFile%;



View(App(3)).WindowVisible(0);





'---------Declare global variables-----------

var newFile%;									'Handle for resampled data file

var oldSize;									'Bin size in source file

var newSize;									'Bin size in new file

var points%;									'Number of data points in source file frames

var newPoints%;									'Number of points in resampled file

var factor;									'Re-sampling factor	

var numChans%;									'Number of channels

var chan%,frame%;								'Counters for loops

var numFrames%;									'Number of frames





'---------Set toolbar buttons--------------

ToolbarSet(1,"Quit");

ToolbarSet(3,"Load source file",Load%);

ToolbarSet(5,"Resample",DoReSample%);

ToolbarSet(7,"Save",DoSave%);

ToolbarEnable(5,0);

ToolbarEnable(7,0);

Toolbar("Downsampling script",1023);





Func Load%()

if oldFile% > 0 then 							'If existing file opened

	View(oldFile%);

	FileClose(-1);								'Close existing file	

endif;

oldFile%:=FileOpen("",0,0);						'Open new source file

if oldFile% < 0 then							'Check file opened

	Message ("Failed to open file");

endif;

Window(0,0,100,100);							'Position window

FrontView(oldFile%);							'Bring to front

if oldFile% >0 then

	ToolbarEnable(5,1);

endif;

return 1;

end;





Func DoReSample%()									

DlgCreate("Re-sampling factor");				'Dialog to enter Resampling factor

DlgReal(1,"Enter factor for re-sampling",0.001,100);

DlgShow(factor);

View(oldFile%);

points%:=Len(View(oldFile%,1).[]);				'Get number of data points in source file

numChans%:=ChanCount(-4);						'Count waveform channels

numFrames%:=FrameCount();							

newPoints%:=points%*factor;					'Get number of data points for new file

oldSize:=Binsize(1);							'Get bin sizes for files

newSize:=oldSize / factor;



newFile%:=SetMemory(numChans%,newPoints%,newSize,View(oldFile%).MinTime(),0,0,0,"Re-sampled","s");	

												'Create empty data file with adjusted bin size and length 

Window(50,0,100,100);

ReSample();

return 1;

end;





Proc ReSample()								'Procedure to loop through frames and channels

var countF%;									'Counter to loop through each frame

var countC%;									'Counter to loop through each channel

for countF%:=1 to numFrames% do					'For each frame

	View(oldFile%).Frame(countF%);				'Ensure correct frame is viewed	

	View(newFile%).Frame(countF%);

	for countC%:=1 to numChans% do				'For each channel	

		DSChan(countC%);						'Link to procedure for downsampling channel

	next;

	if countF% < numFrames% then				'if not the last frame

		View(newFile%).AppendFrame();			'Append empty frame to new file	

	endif;

next;

View(oldFile%);Window(0,0,50,100);				'Position window

FrontView(oldFile%);					

FrontView(newFile%);

ToolbarEnable(7,1);

end;





Proc DSChan(chan%)

var count%;											'Counter to loop through data points

var num%:=0;

var oldDataArr[points%];					'Declare array to contain data from source channel	

ArrConst(oldDataArr[],View(oldFile%,chan%).[]);		'Put data into array

var newDataArr[newPoints%];						'Declare array for new data

'PrintLog("Old %d new %d\n", points%, newPoints%);	'For debug purposes

ArrSpline(newDataArr[],oldDataArr[],1/factor);

ArrConst(View(newFile%,chan%).[],newDataArr[]);		'Copy array to new file

View(newFile%).Optimise(-1);						'Optimise data 

end;



Func DoSave%()

var goodSave%;

View(newFile%);

goodSave%:=FileSaveAs("",0,"Save new file");

if not goodSave% then ToolbarEnable(7,0)endif;

return 1;

end;

david

'$SampleTime.s2s|allows the user to view actual time in hh:mm:ss format at the current time during sampling



'The script is a 'work in progress' and is offered without guarantees.

'You must test it to see whether it is suitable for your application.



var h%;										'Hours

var m%;										'minutes	

var s%;										'seconds

var hTime,mTime,sTime;						

var curTime$;								'String variable to hold time in hh:mm:ss format

var sTime$;									'start time in string format as read from file comment 

var sh$,sh%;								'hours from start time

var sm$,sm%;								'minutes "      "

var ss$,ss%;									'seconds "      "

var lTime,nTime;								'Previous and current position of cursor

var data%;									'Data file





ToolbarSet(1,"Quit");						'Set toolbar	

Toolbarset(3,"New file",New%);

ToolbarSet(5, "Start", Start%);

ToolbarEnable(5,0);

Toolbar("",1023);



Func New%()								'Function to open file and set up cursor etc

data%:=FileNew(0,0);					'Open data file

lTime := 0;

View(data%);	

sh%:=0;

sm%:=0;

ss%:=0;

Window(0,0,100,100);					'Position window and bring to front

FrontView(data%);

ToolbarEnable(3,0);

ToolbarEnable(5,1);

ToolbarSet(0,"",Idle%);				'Start idle routine

return 1;

end;







Func Start%()

View(data%);

SampleStart();               'Start sampling

ToolbarSet(5, "Stop", Stop%);

Return 1;

end;



Func Stop%()

View(data%);

SampleStop();               'Stop sampling

ToolbarSet(5,"Start", Start%);

ToolbarEnable(3,1);

ToolbarEnable(5,0);

return 1;

end;



Func Idle%()							'Idle routine to calculate cursor time and update

View(data%);

nTime:=MaxTime();					'get current time

if nTime <> lTime then				'If cursor time has changed

	hTime:=nTime/3600;			'Get number of hours	

	h%:=Trunc(hTime);							

	mTime:=(hTime-h%)*3600;		'Subtract number of hours and multiply to find minutes remaining

	mTime:=mTime/60;				'Get number of minutes	

	m%:=Trunc(mTime);

	sTime:=(mTime-m%)*60;		'Subtract minutes to find remaining seconds

	s%:=sTime;														

	s%:=s%+ss%;					'Add seconds to start time

	if s% > 59 then 					'If more than 1 minute, add minute and get remaining seconds

		s%:=s%-60;

		m%+=1;

	endif;

	m%:=m%+sm%;					'Add minutes to start time

	if m% > 59 then

		m%:=m%-60;

		h%+=1;

	endif;											

	h%:=h% + sh%;					'Add hours to start time		

	if h% > 23 then

		h%:=h%-24;

	endif;

	curTime$:=Print$("%.2d:%.2d:%.2d",h%,m%,s%);		'Put cursor time in string format

	ToolbarText(curTime$);									    'Write time to toolbar

	lTime:=nTime;												'set lTime to current time

endif;

return 1;

end;

